
1. Method  

1.1 Time-stratified propensity score matching  

We conducted a time-stratified propensity score (PS) matched cohort study to 

examine the relation of allopurinol initiation to the risk of all-cause mortality. PS was 

estimated for the eligible participants based on the covariates preceding their index 

dates and were used to match each allopurinol initiator to a contemporaneous non-

initiator. To adjust for potential secular trends in allopurinol use and mortality, we 

divided the study period (1 January 2000–1 January 2018) into 18 1-year cohort 

accrual blocks. Within each annual cohort accrual block, allopurinol initiators were 

defined as participants who started allopurinol during that year, while participants 

who did not start allopurinol during the same annual accrual block served as potential 

comparators. The date of the first allopurinol prescription was assigned as the index 

date for allopurinol initiators, and a random date within that time block was assigned 

as the index date for non-initiators. Within each cohort accrual block, we calculated 

PS for initial prescription of allopurinol using logistic regression. The variables 

included in the model were sociodemographic factors (i.e., age, sex, socioeconomic 

deprivation index score, and region), body mass index (BMI), lifestyle factors (i.e., 

smoking status and alcohol consumption), CKD severity, SU, eGFR, comorbidities 

(i.e., congestive heart failure, myocardial infarction, stroke, hypertension, angina, 

diabetes mellitus, hyperlipidemia, chronic obstructive pulmonary disease, ischemic 

heart disease, pneumonia or infection, varicose veins, depression, and lupus) prior to 

the index date, medication use (i.e., antihypertensive drug, statin, antidiabetic drug, 

diuretics, aspirin, systemic corticosteroid, topical corticosteroid, non-steroidal anti-

inflammatory drugs [NSAIDs], nitrates, and colchicine) and healthcare utilization 

during the one year prior to the index date. Within each time block, each eligible 

allopurinol initiator was matched to a contemporaneous non-initiator within a 0.1 

caliper of propensity without replacement (40). The PS greedy matching method 

attempted to achieve the comparability of the potential confounders at baseline 

between allopurinol initiators and non-initiators. A previous study suggested that 

optimal matching may not result in improved balance in measured baseline covariates 

when comparing with greedy match (75). In addition, using Monte Carlo simulations, 

increasing the number of untreated subjects matched to each treated subject, on 

average, increased the bias of the estimated treatment effect; conversely, it tended to 

result in increased precision (76). To minimize potential bias, we only matched one 

non-initiator to each allopurinol initiator with the greedy matching method in the 

primary analysis. To avoid duplicated subjects in the PS-matched cohorts, subjects 

who were PS-matched in the early annual accrual block were excluded from the later 

annual accrual block (i.e., sampling without replacement). Subjects who were not PS-

matched to the initiators in the early annual accrual block were still eligible to serve 

as the potential comparators in the late annual accrual block if they remained non 

allopurinol initiators. Thereby, there were replicates in the pre-matched comparison 

cohort and we used “records” to describe the number of the comparison cohort before 

PS matching. Participants in each cohort were followed until the first of the following 

events: death, disenrollment from a GP practice participating in THIN, five years 

follow-up, or the end of the study (30 April 2019).  

 

Rate difference calculation  

We estimated the absolute rate difference (RD) and its 95% confidence intervals [CI]) 

in mortality between the two comparison groups using the following formula:  

RD = rate (initiators) - rate (non-initiators) 

Where a and b refer to the number of events in each cohort, PTa and PTb refer to the 

total person-time accumulated in each cohort, and 95% CI: RD ± 1.96*SERD.  

 

Cumulative mortality curve  

The cumulative mortality curve over 5 years of follow-up plots the inverses of 

Kaplan-Meier estimates, which is a non-parametric statistic used to estimate the 

survival function from lifetime data.  

 

Hazard ratio calculation  

We obtained the hazard ratio (HR) of mortality for the allopurinol initiators vs. non-

initiators using the Cox proportional hazard model. The Cox model is expressed by 

the hazard function denoted by h(t). Briefly, the hazard function can be interpreted as 

the risk of dying at time t. It can be estimated as follows:  

ℎ(𝑡) = ℎ0(𝑡) × 𝑒𝑥𝑝(𝑏1𝑥1 + 𝑏2𝑥2 + ⋯ + 𝑏𝑝𝑥𝑝) 

where:  

➢ t represents the survival time;  

➢ h(t) is the hazard function determined by a set of p covariates (x1, x2,…xp); the 

coefficients (b1, b2,…bp) measure the impact (i.e., the effect size) of covariates;  

➢ the term h0 is measures the baseline hazard. It corresponds to the value of the 

hazard if all the xi are equal to zero (the quantity exp(0) equals 1).  

The exponentiate of bi, i.e., exp(bi), is a HR. In the PS-matched cohort study, we 

only included the indicator of the allopurinol initiator (i.e., allopurinol initiator vs. 

non-initiator) in the model because the baseline characteristics were well-balanced 

after PS matching. To obtain appropriate standard error (SE) and confidence interval 

of HR in PS-matched study design, we performed a “sandwich estimation” in the Cox 

proportional hazards model, accomplished in SAS PROC PHREG.  

 

1.2 Target trial emulation  

To assess the effect of achieving target SU level with allopurinol (or allopurinol dose 

escalation) on the risk of mortality, we emulated a hypothetical target trial comparing 

mortality in participants who achieved target SU level (or who escalated their 

allopurinol dose) within one year after the initiation of allopurinol with that in 

participants who did not achieve target SU level (or who did not escalate allopurinol 

dose) within one year after initiation of allopurinol. Because the exposure of interest 

(i.e., time to “achieving target SU level” or “allopurinol dose escalation”) was 

“duration to reach an exposure level”, we adopted a “cloning, censoring, and 

weighting” approach that emulated an RCT to test the hypotheses using observational 

data (41-43). This approach aims to eliminate immortal time bias in the estimates of 

absolute and relative risk (42). 

 

Cloning: assign patients to a treatment strategy at time zero  

We assigned each person to two treatment strategies (i.e., “achieving target SU level 

vs. “not achieving target SU level” or “allopurinol dose escalation” vs. “no allopurinol 

dose escalation” within one year after initiation of allopurinol). Assigning a person to 

two treatment strategies simultaneously is equivalent to having two copies (or clones) 

of the person in the dataset, with each copy assigned to a different strategy. For 

example, when assessing the effect of “achieving target SU level” during one year 

after the index date on the risk of mortality, we assigned one clone to the “achieving 

target SU level” arm (i.e., achieving the target SU during one year after the index 

date, A=1) and the other clone to the “not achieving target SU level” arm (i.e., not 

achieving the target SU during one year after the index date, A=0). “Cloning” makes 

two comparison groups compatible with their observed data at time zero.  

 

Censoring: ensure that people follow their assigned strategy during the first year 

follow up  

We allowed for a grace period of one year measured from baseline for individuals to 

achieve the target SU level or to escalate the allopurinol dose (14, 44). If clones 

deviated from their assigned strategy during the first year of follow-up, they would be 

artificially censored. Specifically, clones assigned to the “achieving target SU level” 

arm (or “allopurinol dose escalation”) (A=1) were censored if they did not achieve the 

target SU at the end of the first year of follow-up, and clones assigned to the “not 

achieving target SU level” (or no allopurinol dose escalation) arm (A=0) were 

censored if they achieved the target SU (or increased allopurinol dose) at during the 

first year of follow-up. During the grace period, if an individual died before achieving 

the target SU or escalating allopurinol dose, that person was considered consistent 

with his/her assignment in both arms (or clones) and contributed the death outcome to 

each of the assigned arms.  

 

Weighting: adjust for selection bias  

To eliminate the selection bias due to artificial censoring, we used inverse probability 

weighting (42). Specifically, we used the baseline covariates to estimate the 

probability of being artificially censored at the end of the first year of follow-up and 

build the inverse-probability weight as below:  

𝑓(𝐴𝑡|𝐴𝑡−1, 𝐿0, 𝐷𝑡−1 = 0) = 𝜃0 + 𝜃1𝐿0 

𝑊 =
1

𝑓(𝐴𝑡|𝐴𝑡−1, 𝐿0, 𝐷𝑡−1 = 0)
 

where 𝐴 is an indicator for treatment arm, 𝐿0 is a vector of baseline covariates, D is 

an indicator for death. The denominator of the IPW was the probability that a replicate 

adhered to his/her assigned intervention arm using the logistic regression model 

which consisted of the baseline covariates (see Assessment of covariates). Since we 

gave clones one year to achieve the target SU (or to escalate allopurinol dose), clones 

assigned to A=1 are not artificially censored if they died during the first year, because 

they adhered to their strategy. That is, the probability of being uncensored is 1 for 

clones who assigned to A=1 and died during the first year. The inverse probability 

weighting allowed us to create a pseudo-population in which no one is censored, and 

everybody follows their assigned treatment strategy.  

 

Estimating weighted hazard ratio:  

We used a pooled logistic regression model to estimate the hazard as below:  

𝑙𝑜𝑔𝑖𝑡[𝑃𝑟 (𝑌𝑡+1 = 1|𝑌𝑡 = 0, 𝐴, 𝐿0)] = 𝛼0,𝑡 + 𝛼1𝐴 + 𝑎2𝐿0 

where 𝑌𝑡+1 is an indicator (1: yes, 0: no) for death at year 𝑡 + 1, A is an indicator for 

treatment arm. The pooled logistic regression model included an indicator for the 

treatment strategies (A), the baseline covariates (L0), and the year of follow-up (linear 

and quadratic term) (52, 53).  

 

Estimating absolute risk difference:  

We fitted a pooled logistic model that included a product term between the treatment 

arm indicator and time, and estimated the predicted survival probability at time t for 

individual i under regime 𝑎,  𝑆̂𝑖,𝑡
𝑎  , conditional on the individual’s baseline values of 

confounders 𝐿0:  

𝑙𝑜𝑔𝑖𝑡[𝑃𝑟(𝑌𝑡 = 1|𝑌̅𝑡−1 = 0 , 𝐴, 𝐿0)] = 𝛼0,𝑡 + 𝛼1𝐴 + 𝛼2𝐴 ∗ 𝑡 + 𝛼3
𝑇𝐿0 

𝑆̂𝑖,𝑡
𝑎 = ∏[1 −

𝑒𝑥𝑝(𝛼0,𝑘 + 𝛼1𝐴 +  𝛼2𝐴 ∗ 𝑘 + 𝛼3
𝑇𝐿0)

1 + 𝑒𝑥𝑝(𝛼0,𝑘 + 𝛼1𝐴 +  𝛼2𝐴 ∗ 𝑘 + 𝛼3
𝑇𝐿0)

𝑡

𝑘=1

] 

We then used the empirical distribution of the baseline confounders in the entire 

study population to calculate standardized survival probabilities at each time-point:  

𝑆̂𝑡
𝑎 =

1

𝑛
∑ 𝑆̂𝑖,𝑡

𝑎

𝑛

𝑖=1

 

where n is the total number of individuals. Risks were calculated by subtracting the 

survival probabilities from 1 at time t. We estimated the confidence interval of the risk 

differences at 5 years of follow-up using nonparametric bootstrapping with 100 

samples (54).   

2. SAS code for the PS matching analysis and target trial emulation 

analysis  

/*greedy PS matching*/ 

/*Choosing subjects from the first time-block*/ 

%let maxblock=18; 

data pat; 

        set allo; 

  if block=1; 

 run; 

 

/*Calculating PS*/ 

PROC LOGISTIC DATA=pat Descend noprint; 

class treat(ref="0") drink_status (ref="0") 

smoking_status_ordinal(ref="0") sex(ref="2") sdin(ref="1") country 

(ref="E")/param=ref; 

MODEL treat= bmi diabete drink_status hypertension mi stroke angina 

varicose pneu_infl hyperlipid depression tia chf COPD ihd 

smoking_status_ordinal gfr ua hospitalization visit_n refer sdin sex 

age country ckd_stage lupus colchicine ACE antico asp antidiab BETA 

CCBs cortico_syst cortico_topical loop nitrate nsaids potsparing 

statin thiazides; 

OUTPUT OUT= Propen1 prob=prob; run; 

 

/*PS Greedy Matching*/ 

proc psmatch data=Propen1 region=cs; 

class treat country; 

psdata ps=prob treatvar=treat(treated='1'); 

match method=greedy (k=1 order=random(seed=123)) distance=lps 

caliper=0.1 exact=country; 

output out(obs=match)=psmatched1 lps=_Lps matchid=_MatchID; 

run; 

 

data ps_match; 

set psmatched1; 

run; 

 

/*Starting the loop from the second time-block to the last time-

block*/ 

%macro runps; 

%do iloop=2 %to &maxblock; 

 

/*Removing the PS-matched subjects in early accrual time blocks from 

the later accrual time blocks*/ 

proc sql; create table allo1_1 as  

 select * from allo where subjid not in (select subjid from 

ps_match); 

quit; 

 data pat&iloop; 

        set allo1_1; 

  if block=&iloop; 

 run; 

PROC LOGISTIC DATA=pat&iloop Descend noprint; 

class treat(ref="0") drink_status (ref="0") 

smoking_status_ordinal(ref="0") sex(ref="2") sdin(ref="1") country 

(ref="E")/param=ref; 

MODEL treat= bmi diabete drink_status hypertension mi stroke angina 

varicose pneu_infl hyperlipid depression tia chf COPD ihd 

smoking_status_ordinal gfr ua hospitalization visit_n refer sdin sex 

age country ckd_stage lupus colchicine ACE antico asp antidiab BETA 

CCBs cortico_syst cortico_topical loop nitrate nsaids potsparing 

statin thiazides; 

OUTPUT OUT= Propen&iloop prob=prob; run; 

 

proc psmatch data=Propen&iloop region=cs; 

class treat country; 

psdata ps=prob treatvar=treat(treated='1'); 

match method=greedy (k=1 order=random(seed=123)) distance=lps 

caliper=0.1 exact=country; 

output out(obs=match)=psmatched&iloop lps=_Lps matchid=_MatchID; 

run; 

 

/*Combining all PS matched dataset*/ 

data ps_match; 

set Psmatched1-psmatched&iloop; 

run; 

 

%end; 

%mend; 

%runps; 

 

 

data ps_match_greedy; 

set ps_match; 

run; 

 

/*Cox proportional model*/ 

/*Sandwich estimation*/ 

proc phreg data=ps_match_greedy COVSANDWICH; 

class treat(ref='0')_matchid/param=ref; 

model followy*death(0)=treat/risklimits; 

id _matchid; 

run; 

/*Target trial emulation */ 

/*data d_all represents the person-year format data of 

allopurinol initiators*/ 

/*A=1 represents that individual actually achieve the target 

SU during the first year after the index date*/ 

/*A=0 represents that individual actually does not achieve the 

target SU during the first year after the index date*/ 

 

/* STEP 1: Create replicates (clones) for each individual */ 

data pooled_TTE3; 

set d_all; 

treat=0; output; 

treat=1; output; 

run; 

 

/* STEP 2: Censor replicates when they deviate from their assigned 

strategy */ 

proc sort data=pooled_TTE3; by subjid treat time_b; run; 

data pooled_TTE3; 

set pooled_TTE3; 

by subjid treat time_b; 

retain c; 

if first.treat then c=0; 

if treat=0 and A=1 then c=1; 

if treat=1 and A=0 then c=1; 

 

if c=0 then adhr=1; 

if c=1 then adhr=0; 

run; 

 

data nocensor censor; 

set pooled_TTE3; 

if c=0 then output nocensor; 

if c=1 then output censor; 

run; 

 

proc sort data=censor; by subjid treat time_b; run; 

data censor; 

set censor; 

by subjid treat time_b; 

if first.treat; 

run; 

 

data pooled_TTE4; 

set nocensor censor; 

run; 

 

/*STEP 3: Estimate probabilities for denominator of the IP weights */ 

proc logistic data =pooled_TTE4 descending; 

class adhr(ref="0") drink_status (ref="0") 

smoking_status_ordinal(ref="0") sex(ref="2") sdin(ref="1") country 

(ref="E")/param=ref; 

MODEL adhr=bmi diabetes drink_status hypertension mi stroke angina 

varicose pneu_infl hyperlipid depression tia chf COPD ihd 

smoking_status_ordinal gfr first_ua hospitalization visit_n refer 

sdin sex age country ckd_stage colchicine ACE antico asp antidiab 

BETA CCBs cortico_syst cortico_topical loop nitrate nsaids potsparing 

statin thiazides; 

output out = _modelAd_1 (keep = subjid pdenom_1 time_b) p=pdenom_1; 

where time_b=0 and treat=1; 

run; 

 

proc logistic data =pooled_TTE4 descending; 

class adhr(ref="0") drink_status (ref="0") 

smoking_status_ordinal(ref="0") sex(ref="2") sdin(ref="1") country 

(ref="E")/param=ref; 

MODEL adhr= bmi diabetes drink_status hypertension mi stroke angina 

varicose pneu_infl hyperlipid depression tia chf COPD ihd 

smoking_status_ordinal gfr first_ua hospitalization visit_n refer 

sdin sex age country ckd_stage colchicine ACE antico asp antidiab 

BETA CCBs cortico_syst cortico_topical  loop nitrate nsaids 

potsparing statin thiazides; 

output out = _modelAd_0 (keep = subjid pdenom_0 time_b) p=pdenom_0; 

where time_b=0 and treat=0; 

run; 

 

/*  STEP 4: Build the unstabilized IP weights */ 

data pooled_TTE4_1 pooled_TTE4_0; 

set pooled_TTE4; 

if treat=1 then output pooled_TTE4_1; 

if treat=0 then output pooled_TTE4_0; 

run; 

 

/*  STEP 4_1 Build the weights for subjects who were assigned to 

achieving the target SU during the first year */ 

proc sql; 

create table pooled_TTE4_11 as 

select a.*, b.pdenom_1 from pooled_TTE4_1 a left join _modelAd_1 b 

on a.subjid=b.subjid; 

quit; 

 

data pooled_TTE4_11; 

set pooled_TTE4_11; 

unsw = 1.0/pdenom_1; 

run; 

 

/* weight set to 1 for subjects who were assigned to achieving the 

target SU during the first year */ 

/* subjects were censored if they did not achieve the target SU level 

at the end of the first year*/ 

data pooled_TTE4_11; 

set pooled_TTE4_11; 

if time_b=0 then unsw=1; 

else if c=1 then unsw=0; run; 

 

/* STEP 4_2 Build the weights for subjects who were assigned to not 

achieving the target SU during the first year */ 

proc sql; 

create table pooled_TTE4_00 as 

select a.*, b.pdenom_0 from pooled_TTE4_0 a left join _modelAd_0 b 

on a.subjid=b.subjid; 

quit; 

 

data pooled_TTE4_00; 

set pooled_TTE4_00; 

unsw = 1.0/pdenom_0; 

run; 

 

/* subjects were censored if they achieved the target SU level*/ 

 

data pooled_TTE4_00; 

set pooled_TTE4_00; 

if c=1 then unsw=0; run; 

 

data pooled_TTE5; 

set pooled_TTE4_11 pooled_TTE4_00; 

run; 

 

data pooled_TTE5; 

set pooled_TTE5; 

if unsw=0 then delete; 

run; 

/* STEP 5: Truncate the weights at the 99th percentile*/ 

 

proc means data = pooled_TTE5 p99 noprint; 

var unsw; 

output out=pctl (keep=p99) p99=p99; 

run; 

data temp; 

set pctl; 

call symput ('cutoff', p99); 

run; 

data pooled_TTE6; 

set pooled_TTE5; 

unsw_t = unsw; 

if unsw>%sysevalf(&cutoff) then do; 

 unsw_t = %sysevalf(&cutoff); 

end; 

run; 

 

proc univariate data=pooled_TTE6; var unsw unsw_t; run; 

 

/*STEP 6: Estimate weighted hazard ratio*/ 

/*Y refers to death*/ 

 

proc genmod data=pooled_TTE6 descending; 

class subjid Y(ref="0") treat(ref="0") drink_status (ref="0") 

smoking_status_ordinal(ref="0") sex(ref="2") sdin(ref="1") country 

(ref="E")/param=ref; 

MODEL Y=treat time_b time_b*time_b bmi diabete drink_status 

hypertension mi stroke depression tia chf COPD ihd angina varicose 

pneu_infl hyperlipid smoking_status_ordinal gfr first_ua 

hospitalization visit_n refer sdin sex age country ckd_stage 

colchicine ACE antico asp antidiab BETA CCBs cortico_syst 

cortico_topical loop nitrate nsaids potsparing statin thiazides; 

scwgt unsw_t; 

repeated subject=subjid/type=ind; 

estimate "Achieving target SU" treat 1/exp; 

run; 

 

/*Calculating the five-year risk difference and plotting the weighted 

survival curves*/ 

 

data pooled_TTE6; 

set pooled_TTE6; 

time_b2=time_b*time_b; 

randtime_b=treat*time_b; 

randtime_b2=treat*time_b2; 

run; 

 

proc genmod data=pooled_TTE6 descending; 

class subjid treat(ref="0")/param=ref; 

model Y=treat time_b time_b2 randtime_b randtime_b2/link=logit 

dist=bin;  

weight unsw_t; 

ods output ParameterEstimates = plrFit_ix_SWT; 

run; 

 

data plrFit_ix_SWT; 

set plrFit_ix_SWT; 

 if PARAMETER = "Scale" then delete; 

run; 

 

proc sql noprint; 

 select ESTIMATE FORMAT =16.12 INTO: IBC_ESTIMATE separated by ' ' 

from plrFit_ix_SWT; 

quit; 

 

proc sql noprint; 

 select PARAMETER INTO: MODEL separated by ' ' from plrFit_ix_SWT; 

quit; 

proc means sum noprint data = plrFit_ix_SWT;  

 var DF; 

 output out = nobs (drop = _type_ _freq_ where=(_stat_ ="N")); 

run; 

 

proc sql noprint;  

 select DF into:NVAR separated by ' ' from nobs; quit; 

 

data untreated (keep = s ci treat time_b); 

set pooled_TTE6; 

where time_b = 0; 

 array var{&nvar} &model; 

 array coef{&nvar} (&ibc_estimate); 

 intercept = 1; 

 s=1; 

 treat = 0; 

 do time_b = 0 to 4; 

  time_b2 = time_b*time_b; 

  xbeta = 0; 

     randtime_b=treat*time_b; 

        randtime_b2=treat*time_b2; 

  

  do i = 1 to dim(var); 

   xbeta = xbeta + coef[i] *var[i]; 

  end; 

     p = 1/(1+exp(-xbeta)); 

  s = s*(1-p); 

  ci = 1-s; 

  output; 

 end; 

run; 

 

data treated (keep = s ci treat time_b); 

set pooled_TTE6; 

where time_b = 0; 

 array var{&nvar} &model; 

 array coef{&nvar} (&ibc_estimate); 

 intercept = 1; 

 s=1; 

 treat = 1 ; 

 do time_b = 0 to 4; 

  time_b2 = time_b*time_b; 

  xbeta = 0; 

     randtime_b=treat*time_b; 

        randtime_b2=treat*time_b2; 

  

  do i = 1 to dim(var); 

   xbeta = xbeta + coef[i] *var[i]; 

  end; 

     p = 1/(1+exp(-xbeta)); 

  s = s*(1-p); 

  ci = 1-s; 

  output; 

 end; 

run; 

 

data both; 

set untreated treated; 

by treat; 

run; 

 

/* Calculate the mean survival at each year within each treatment 

arm*/ 

proc means data = both mean; 

 class time_b treat; 

 types treat*time_b; 

    var s; 

    output out = means (drop = _type_ _freq_) mean(s) = s; 

run; 

 

proc transpose data=means out = wideres prefix = Surv_; 

var s; 

id treat; 

by time_b; 

run; 

 

data wideres; 

set wideres; 

 time_b = time_b+1; 

 merge_id = 1; 

    rd = (1-Surv_1) -(1- Surv_0); 

run; 

 

data wideres; 

set wideres; 

total + ratio; 

drop total ratio; 

run; 

 

data zero; 

 time_b = 0; 

 merge_id = 1; 

 surv_0 = 1; 

 surv_1 = 1; 

 rd = 0; 

run; 

 

data results_1; 

merge wideres zero; 

by time_b; 

drop merge_id _NAME_; 

run; 

 

/*Plot the results*/ 

data results_1; 

set results_1; 

mortality_1=1-surv_1; 

mortality_0=1-surv_0; 

run; 

 

goptions reset=all; 

axis1 length=5.9in label=(angle=90 height=2 "Risk of Death") width=1 

order=(0 to 0.3 by .1)  

     major=(height=2 width=2) minor=none value=(height=1.5) 

offset=(2,2); 

axis2 length=5.9in label=(height=2 justify=center "Years of follow 

up")  

     width=1 order=(0 to 5 by 1) major=(height=1 width=2)  

     minor=none value=(height=1.5) offset=(2,2); 

symbol1 c=red h=12 l=1 w=2 v=none i=line mode=include; 

symbol2 c=blue h=12 l=4 w=2 v=none i=line mode=include; 

legend1 label=none value=(color=black h=1.5 "Achieving target SU 

level" "Not achieving target SU level")  

     down=4 position=(MIDDLE  left inside) FWIDTH=1 

     shape=symbol(6,3) offset=(0,1) mode=protect; 

 

proc gplot data=results_1; plot (mortality_1 

mortality_0)*time_b/overlay vaxis=axis1 haxis=axis2 

     legend=legend1 noframe; 

run;  

quit;  

 

/* Print risk difference*/ 

proc print data = results_1; 

var time_b mortality_1 mortality_0 rd; 

run; 

 


